Population Effect Estimation in Bayesian Hierarchical Models Using Sum-to-Zero Constraints

StanCon 2026

Sean Pinkney
sean.pinkney@gmail.com
SVP Measurement Analytics at Publicis Groupe
Stan Developer

Nikolai Vetr
nikgvetr@stanford.edu
Department of Pathology, Stanford University

Slides and code at

github.com/spinkney/open-talks

Agenda

- Preliminaries
- What is a sum-to-zero constraint?
- Hierarchical models between frequentist and bayesian approaches
- How is it possible to translate between them?
- Implications

Preliminaries: Distributions

We will restrict ourselves to elliptically symmetric distributions for hierarchical models, i.e.,

$$f(x) = c\, g(x^\top \Sigma_x^{-1} x),$$
$$\int_{\mathbb{R}^d} f(x)\, dx = 1$$

where[1]

- the generator $g(\cdot) : \mathbb{R}^+ \to \mathbb{R}^+$
- the normalizing constant c ensures $f(x)$ integrates to 1

Preliminaries: Distributions

$$f(x) = c\, g(x^\top \Sigma_x^{-1} x),$$
$$\int_{\mathbb{R}^d} f(x)\, dx = 1$$

The choice of g controls the tail behavior, allowing both light- and heavy-tailed priors within the same family:

- Normal: $g(t) \propto e^{-t/2}$
- Student-t_ν (Cauchy at $\nu = 1$): $g(t) \propto (1 + t/\nu)^{-(\nu+d)/2}$
- Logistic: $g(t) \propto e^{-t} / \left(1 + e^{-t}\right)^2$
- Laplace: $g(t) \propto e^{-\sqrt{t}}$

What is a sum-to-zero constraint?

It's simply

$$\sum_{j}^{n} a_j = 0$$

To construct one might randomly draw i.i.d. $n - 1$ values y and

$$y_n \stackrel{\text{set}}{=} -\sum_{i}^{n-1} y_i$$

$$\boldsymbol{z} = [y_1 \; y_2 \; \ldots \; y_{n-1} \; y_n]$$

$$\sum_{i}^{n} z = 0$$

What is a sum-to-zero constraint?

Since only z_n is a function of y the marginal density will be the density of the sum of $z_{i:n-1}$. Most modelers would not want this!



The implicit prior on the last element is clearly different. Furthermore, the covariance of the final element is coupled strongly with the other elements.

What is a sum-to-zero constraint?

We can balance the density across the values by using an orthonormal basis for the zero-sum subspace, $QQ^\top = P_0 = I_n - \frac{1}{n}\mathbf{1}_n\mathbf{1}_n^\top$.



All the values are now isotropic and exchangeable.

Preliminaries: Background on Bayesian hierarchical models



Sharing of information happens

- Φ population distribution
- Globally
- Bi-directionally
- AKA **partial pooling**

When the evidence or data for a parameter are

- low
  - → estimate is closer to prior
- large
  - → data swamps prior
  - → prior pull is weak

Preliminaries: Background on frequentist hierarchical models



No population distribution Φ

- the θ_j are **fixed unknowns**, not draws
- no pooling: each θ_j is estimated from group j's data alone
- information flows one way, $\theta_j \to y_j$

Groups are tied together only by a *hard* constraint

- e.g. effects coding $\sum_j \theta_j = 0$
- identifies the grand mean instead of a population parameter

What is the population effect?

Let's take a 2-level, varying slopes, varying intercept model with i units and j groups

$$y_{ij} = \underbrace{\overset{\text{population parameters}}{\alpha} + a_j}_{\text{varying intercept}} + X(\underbrace{\beta + b_j}_{\text{varying slope}}) + \epsilon_{ij}$$

Group-level (conditional): $E(y_{ij} \mid X, j) = \alpha + a_j + X(\beta + b_j)$

Population-level (marginal): $E(y_{ij} \mid X) = \alpha + X\beta$
when $E(a_j) = E(b_j) = 0$

Different populations

The data inform the group coefficients α_j and β_j whereas the population α and β sit one level up.

Bayesian hierarchical model

α and β are *population* parameters and the observed groups are **draws from** that population

$$a_j \sim \mathcal{N}(\alpha, \sigma_a), \quad b_j \sim \mathcal{N}(\beta, \sigma_b)$$

A benefit is that new groups may still be infered from the population distribution as any finite draw of J groups has

$$\bar{a} = \tfrac{1}{J} \sum_j a_j \neq \alpha$$

Frequentist effects coding

No population to draw from — the J observed groups **are** the population

The "population" parameter is the finite-population grand mean

$$\alpha = \tfrac{1}{J} \sum_j (\alpha + a_j) \iff \sum_j a_j = 0$$

Is there an equivalence?

Suppose we restrict to varying intercepts, $x_i \sim \mathcal{N}(\alpha + a_j, \sigma_x)$:

Bayesian hierarchical model

$$a_j^{\text{bayes}} \sim \mathcal{N}(0, \sigma_a)$$

$$\bar{a} = \frac{1}{J} \sum_j a_j^{\text{bayes}} \sim \mathcal{N}\left(0, \frac{\sigma_a^2}{J}\right)$$

Frequentist effects coding

$$\alpha = \tfrac{1}{J} \textstyle\sum_j \left(\alpha + a_j^{\text{s2z}}\right) \iff \textstyle\sum_j a_j^{\text{s2z}} = 0$$

Take

$$\alpha_{\text{s2z}} \overset{\text{set}}{=} \alpha_{\text{bayes}} + \bar{a}$$

$$a_j^{\text{s2z}} \overset{\text{set}}{=} a_j^{\text{bayes}} - \bar{a}$$

$$\sum a_j^{\text{s2z}} = 0$$

Does it work?

Let's simulate six deliberately unbalanced groups from

$$x_i \sim \mathcal{N}(\alpha + a_{j[i]}, \sigma_x), \qquad a_j \sim \mathcal{N}(0, \tau^2),$$

with

```
alpha = 5
tau = 2.5
sigma_x = 1
n_j = (8, 13, 21, 34, 55, 89)
```

For this finite draw,

$$\frac{1}{J}\sum_j a_j^{\text{bayes}} = \bar{a} = 0.85, \qquad \alpha_{\text{s2z}} = 5.85.$$

Sampling 10k iterations over 4 chains. The first Stan program samples unconstrained Bayesian varying effects; the second samples a

Conversion

Bayesian specification and conversion to sum-to-zero equivalent

```
data {
  int<lower=1> N;
  int<lower=2> J;
  array[N] int<lower=1, upper=J> group;
  vector[N] x;
  real<lower=0> alpha_prior_sd;
}
parameters {
  real alpha_bayes;
  vector[J] a_bayes;
  real<lower=0> tau;
  real<lower=0> sigma_x;
}
model {
  alpha_bayes ~ normal(0, alpha_prior_sd);
  a_bayes ~ normal(0, tau);
  tau ~ normal(0, 5);
  sigma_x ~ normal(0, 2);
  x ~ normal(alpha_bayes + a_bayes[group], sigma_x);
}
generated quantities {
  real mean_a = mean(a_bayes);
  real alpha_s2z_recovered = alpha_bayes + mean_a;
  vector[J] a_s2z_recovered = a_bayes - mean_a;
}
```

Conversion

Sum-to-zero specification and conversion to Bayesian equivalent

The two fits agree



Across α_{s2z} and the six a_j^{s2z}, the largest posterior-mean difference is 0.002. Across every plotted parameter, the largest difference is 0.003: Monte Carlo error at the scale of the plot.

What about going from sum-to-zero to population models?

Let s_α denote the prior SD of α_{bayes} and set $v_m = \tau^2/J$. Because $\alpha_{\text{s2z}} = \alpha_{\text{bayes}} + \bar{a}$,

$$w = \frac{v_m}{s_\alpha^2 + v_m}, \tag{1}$$

$$v_c = \frac{s_\alpha^2 v_m}{s_\alpha^2 + v_m}, \tag{2}$$

$$\bar{a} \mid \alpha_{\text{s2z}}, \tau \sim \mathcal{N}(w\, \alpha_{\text{s2z}},\ v_c)\,. \tag{3}$$

For every sum-to-zero posterior draw:

1. Sample $m \sim \mathcal{N}(w\, \alpha_{\text{s2z}}, v_c)$, using (mean, variance) notation.
2. Set $\alpha_{\text{bayes}} = \alpha_{\text{s2z}} - m$ and $a_g^{\text{bayes}} = a_g^{\text{s2z}} + m$.

Sum-to-zero specification and conversion

```
generated quantities {
  real mean_a_bayes;
  real alpha_bayes_recovered;
  vector[J] a_bayes_recovered;
  {
    real var_mean_a = square(tau) / J;
    real var_alpha = square(alpha_prior_sd);
    real weight = var_mean_a / (var_alpha + var_mean_a);
    real conditional_sd = sqrt(
      var_alpha * var_mean_a / (var_alpha + var_mean_a)
    );

    mean_a_bayes = normal_rng(weight * alpha_s2z, conditional_sd);
  }
  alpha_bayes_recovered = alpha_s2z - mean_a_bayes;
  a_bayes_recovered = a_s2z + mean_a_bayes;
}
```

Back to Bayesian: the two fits agree



Across α_{bayes} and the six a_j^{bayes}, the largest posterior-mean difference is 0.032. Across every plotted parameter, the largest difference is 0.032. The largest standardized gap is 1.6 combined Monte Carlo SEs.

On the difference between two posterior means

For each parameter, we calculate

$$z = \frac{\left|\hat{\mu}_{\text{Bayes}} - \hat{\mu}_{\text{S2Z}}\right|}{\sqrt{\text{MCSE}^2_{\text{Bayes}} + \text{MCSE}^2_{\text{S2Z}}}}.$$

The largest value was 1.6. So the worst observed difference was only 1.6 times the noise expected from running two independent finite MCMC simulations. A standard-normal fluctuation exceeds 1.6 in absolute value about 11% of the time.

ESS and ESS/grad on the two parameterizations

Parameter	Bayesian		Sum-to-zero	
	ESS	ESS/grad	ESS	ESS/grad
α	5,557	0.002	33,461	0.121
a_1	5,719	0.002	39,725	0.144
a_2	5,683	0.002	64,383	0.234
a_3	5,650	0.002	71,686	0.260
a_4	5,588	0.002	56,372	0.204
a_5	5,581	0.002	46,215	0.168
a_6	5,572	0.002	39,120	0.142
τ	11,574	0.004	57,662	0.209
σ_x	16,549	0.006	61,344	0.223

4 chains × 10,000 sampling iterations = 40,000 post-warmup draws, plus 4,000 warmup iterations per chain. Bulk ESS is pooled across chains. ESS/grad divides by the total post-warmup `n_leapfrog__`: Bayesian = 2,796,128; sum-to-zero = 275,672.

Take-away

Our Bayesian hierarchical models are doing something special.

Treating the varying effects as random variables allows specifying a distribution. For elliptical distributions we can convert across the two parameterizations!

Don′t be afraid to use sum-to-zero constraints in your models.

More to come in the paper

- Multiple sets of varying effects
- Varying slopes conversions

Thank you!

github.com/spinkney/open-talks

Delmas, Jean-Pierre. 2024. "Background on Real and Complex Elliptically Symmetric Distributions." In *Elliptically Symmetric Distributions in Signal Processing and Machine Learning*, edited by Jean-Pierre Delmas, Mohammed Nabil El Korso, Stefano Fortunati, and Frédéric Pascal. Springer Nature Switzerland. https://doi.org/10.1007/978-3-031-52116-4_1.